
TIRO ZEIHARRA

higidura horizontala: zuzen eta uniformea
s = so + vox t
vox = kte

higidura bertikala: zuzen eta uniformeki azeleratua
s = so + voy t + 1/2 a t2

v = voy + a t
a = -g  (kasu honetan)

Aurreko lerro hauek irakurri ondoren programa
astuna badirudi ere, erabilpena arina da.

Proiektila eta objektiboa azalduko dira pantailan.
Proiektila beti abiapuntu berean agertuko da eta
objektiboaren posizioa berriz, aleatorioki aldatuko da.
Erabiltzailearen zeregina proiektilaz objektiboa jotzea
da. Horretarako, aldez aurretik, angelua eta abiadura
hautatu beharko  ditu.

40. alean hasitako ildoari jarraituz, tiro zeiharraren
simulazioari heldu diogu oraingoan.

Dakigunez, horizontalarekiko α angeluaz eta v
abiaduraz botatako proiektilaren higidura dugu tiro
zeiharraren adibideetako bat; hemen landuko duguna
hain zuzen ere.

Programaren azalpenari ekin aurretik gogora ditza-
gun, labur bada ere, tiro zeiharraren kontzeptua eta
erabiltzen diren formulak.

Mota honetako higiduratan hasierako abiadura bi
osagaitan deskonposatzen da:

abiadura horizontalean: vox = vo cos α
abiadura bertikalean:     voy = vo sin α

Higiduraren deskonposaketa berriz, ondokoa da:

10   REM TIRO ZEIHARRA
20   KEY OFF:CLS:SCREEN 1,0
30   DIM BON(360)
40   GEH$=”B”
50   WHILE GEH$=”B”
60      CLS
70      LOCATE 2,14:PRINT “TIRO ZEIHARRA”:LOCATE 3,14:PRINT “=============”
80      LINE (30,124)-(290,128),1,B:PAINT (32,126),1,1
90      CIRCLE (35,119),5,2:PAINT (35,119),2,2
100     GET (30,114)-(50,123),BON
110     RANDOMIZE TIMER
120     S=INT(RND*110)+100:LOCATE 6,30:PRINT “s =”;S;”m”
130     LINE (30+S,119)-(40+S,124),3,B:PAINT (S+32,121),3,3
140     LINE (41,119)-(46,119):LINE -(43,117):LINE (46,119)-(43,121)
150     LOCATE 19,10:COLOR 0,3:PRINT “angelua”:LINE (72,156)-(128,172),2,B
160     LOCATE 21,8:PRINT CHR$(27):LOCATE 21,18:PRINT CHR$(26)
170     LOCATE 24,10:PRINT “Hautatzeko: <RETURN>”;
180     GRAD$=” 0x30x45x60x90x”:PI=3.141593
190     AN=1:LOCATE 21,12:PRINT “ 0x”:TE$=” “
200     WHILE ASC(TE$)<>13
210        TE$=””:WHILE TE$=””:TE$=INKEY$:WEND
220        WHILE ASC(TE$)<>13 AND ASC(RIGHT$(TE$,1))<>75 AND ASC(RIGHT$(TE$,1))<>77
230           TE$=INKEY$
240           IF TE$=”” THEN TE$=” “
250        WEND
260        A=ASC(RIGHT$(TE$,1))
270        LOCATE 21,12
280        IF A<>75 THEN 300



59

Etxeko ordenadorea
290           IF AN=1 THEN AN=5 ELSE AN=AN-1
300        IF A<>77 THEN 320
310           IF AN=5 THEN AN=1 ELSE AN=AN+1
320        P=(AN-1)*3+1:PRINT MID$(GRAD$,P,3)
330     WEND
340     ANG=VAL(MID$(GRAD$,P,3))*PI/180
350     X1=5*COS(ANG):Y1=5*SIN(ANG):X2=10*COS(ANG):Y2=10*SIN(ANG)
360     LINE (41,119)-(46,119),0:LINE -(43,117),0:LINE (46,119)-(43,121),0
370     LINE (35+X1,119-Y1)-(35+X2,119-Y2):CIRCLE (35+X2,119-Y2),1
380     LOCATE 19,22:COLOR 0,3:PRINT “abiadura”:LINE (168,156)-(232,172),2,B
390     LOCATE 21,20:PRINT CHR$(27):LOCATE 21,31:PRINT CHR$(26)
400     ABI=0:LOCATE 21,24:PRINT “   “:LOCATE 21,24:PRINT ABI
410     TE$=” “
420     WHILE ASC(TE$)<>13
430        TE$=””:WHILE TE$=””:TE$=INKEY$:WEND
440        WHILE ASC(TE$)<>13 AND ASC(RIGHT$(TE$,1))<>75 AND ASC(RIGHT$(TE$,1))<>77
450           TE$=INKEY$
460           IF TE$=”” THEN TE$=” “
470        WEND
480        A=ASC(RIGHT$(TE$,1))
490        LOCATE 21,24
500        IF A<>75 THEN 520
510           IF ABI=0 THEN ABI=46 ELSE ABI=ABI-2
520        IF A<>77 THEN 540
530           IF ABI=46 THEN ABI=0 ELSE ABI=ABI+2
540        PRINT “   “:LOCATE 21,24:PRINT ABI
550     WEND
560     LINE (35+X1,119-Y1)-(35+X2,119-Y2),0:CIRCLE (35+X2,119-Y2),1,0
570     DEN=2*(ABI*SIN (ANG))/10
580     XX=30:YY=114
590     FOR D=0 TO DEN STEP .1
600         PUT (XX,YY),BON
610         X=ABI*COS(ANG)*D
620         Y=ABI*SIN(ANG)*D - 10*D*D/2
630         XX=30+X:YY=114-Y
640         SOUND YY*5,.2
650         PUT (XX,YY),BON
660         FOR I=1 TO 50:NEXT I
670     NEXT D
680     IF X>S-9 AND X<S+9 THEN SOUND 125,10
690     LOCATE 24,1:PRINT SPC(39);
700     LOCATE 24,2:PRINT “Beste saiorik egin nahi al duzu (B/E)?”;
710     GEH$=””:WHILE GEH$<>”B” AND GEH$<>”b” AND GEH$<>”E” AND GEH$<>”e”:GEH$=INKEY$:WEND
720  WEND
730  REM AMAIA TIRO ZEIHARRA

PROGRAMAREN AZALPENA

60-140: Jaurtikiko den proiektila leku jakinean
irudikatzen da. Objektiboa zein distantzia-
tara egongo den aleatorioki kalkulatzen da
eta dagokion lekuan irudikatzen.

150-330: Proiektila zein angelurekin jaurtiki nahi den
hautatu beharko da lehenik. 
Bost angelu aukeratu ahal izango dira: 0°,
30°, 45°, 60° eta 90°. Hautaketa ←, → eta
<RETURN> teklen bidez egin behar da.
Balioak ziklikoki aldatuko dira; hots,
90°ren ondoren 0° azalduko da eta 0°ren
aurretik 90°.

340-370: Hautatutako angelua gezi baten bidez adie-

razten da pantailan.
380-550: Angelua aukeratu ondoren, proiektilari

eman nahi zaion abiadura hautatu behar da. 
0-46 bitarteko balioa izango da eta hau ere
←, → eta <RETURN> teklen bidez eta zi-
klikoki hautatu behar da.

560-670: Hautatutako balioetan oinarrituz, proiek-
tilak berriro lurrera iristeko behar duen
denbora kalkulatzen da (570).
Denbora horrekin zikloa osatzen da. Segun-
do-hamarren bakoitzeko proiektilaren posi-
zioa kalkulatzen da eta pantailan irudika-
tzen.

680: Proiektilak objektiboa jo duen ala ez egiazta-
tzen da.


